Homogeneous Second-Order Differential Equations

James Rohal

We are going to use Maple to solve some second-order differential equations. They take the form $P(x)\frac{d^2y}{dx^2} + Q(x)\frac{dy}{dx} + R(x)y = G(x)$. A homogeneous second-order differential equations has $G(x) = 0$ so it looks like $P(x)\frac{d^2y}{dx^2} + Q(x)\frac{dy}{dx} + R(x)y = 0$. We are going to look at a special case where $P(x) = a$, $Q(x) = b$, $R(x) = c$ and a, b, c are constants.

$y'' + y' - 6y = 0$

Solve the characteristic equation

```
> solve(r^2 + r - 6 = 0, r);
```

$$2, -3 \tag{1.1}$$

Two different real solutions means that the general solution is $y(x) = c_1e^{2x} + c_2e^{-3x}$ where c_1, c_2 are constants. We can verify this using **dsolve**.

```
> dsolve(diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = 0, y(x));
```

$$y(x) = \_C1\, e^{-3x} + \_C2\, e^{2x} \tag{1.2}$$

Maple denotes the constants as $\_C1$ and $\_C2$ rather than c_1 and c_2.

$4y'' + 12y' + 9y = 0$

Solve the characteristic equation

```
> solve(4*r^2 + 12*r + 9 = 0, r);
```

$$-\frac{3}{2}, -\frac{3}{2} \tag{2.1}$$

One real solutions means that the general solution is $y(x) = c_1e^{-\frac{3}{2}x} + c_2xe^{-\frac{3}{2}x}$ where c_1, c_2 are constants. We can verify this using **dsolve**.

```
> dsolve(4*diff(y(x), x$2) + 12*diff(y(x), x) + 9*y(x) = 0, y(x))
  ;
```

$$y(x) = \_C1\, e^{-\frac{3}{2}x} + \_C2\, e^{-\frac{3}{2}x}\, x \tag{2.2}$$

Maple denotes the constants as $\_C1$ and $\_C2$ rather than c_1 and c_2.

$y'' - 6y' + 13y = 0$

Solve the characteristic equation

```
> solve(r^2 - 6*r + 13 = 0, r);
```

$$3 + 2\,I, 3 - 2\,I \tag{3.1}$$

No real solutions means that the general solution is $y(x) = e^{3x}\left(c_1 \cos(2x) + c_2 \sin(2x)\right)$ where c_1, c_2 are constants. We can verify this using **dsolve**.

```
> dsolve(diff(y(x), x$2) - 6*diff(y(x), x) + 13*y(x) = 0, y(x));
```

$$y(x) = \_C1\, e^{3x} \sin(2x) + \_C2\, e^{3x} \cos(2x) \qquad \textbf{(3.2)}$$

Maple denotes the constants as $\_C1$ and $\_C2$ rather than c_1 and c_2.

Initial Value Problem

An IVP has conditions of the form $y(x_0) = y_0$ and $y'(x_0) = y_1$. Let's solve the initial value problem $y'' + y' - 6y = 0, y(0) = 1, y'(0) = 3$.

We already know the general solution is $y(x) = c_1 e^{2x} + c_2 e^{-3x}$ where c_1, c_2 are constants. The conditions allow us to find c_1, c_2. I am going to call the constants **_C1** and **_C2** so that it matches with the notation Maple gives us when using **dsolve**.

```
> y_original := _C1*exp(2*x) + _C2*exp(-3*x);
```

$$y\_original := \_C1\, e^{2x} + \_C2\, e^{-3x} \qquad \textbf{(4.1)}$$

```
> y_prime := diff(y_original,x);
```

$$y\_prime := 2\, \_C1\, e^{2x} - 3\, \_C2\, e^{-3x} \qquad \textbf{(4.2)}$$

To use the conditions, we plug them our expressions for $y(x)$ and $y'(x)$ which are given as **y_original** and **y_prime**. This gives us a system of equations to solve.

```
> eqn1      := 1 = eval(y_original, x=0);
  eqn2      := 3 = eval(y_prime, x=0);
  constants := solve([eqn1, eqn2], {_C1, _C2});
```

$$eqn1 := 1 = \_C1 + \_C2$$

$$eqn2 := 3 = 2\, \_C1 - 3\, \_C2$$

$$constants := \left\{\_C1 = \frac{6}{5}, \_C2 = -\frac{1}{5}\right\} \qquad \textbf{(4.3)}$$

Now that we know the constants **_C1** and **_C2**, we can substitute them back in to our original expression.

```
> eval(y_original, constants);
```

$$\frac{6}{5}\, e^{2x} - \frac{1}{5}\, e^{-3x} \qquad \textbf{(4.4)}$$

This is the same solution that **dsolve** gives us.

```
> dsolve({diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = 0, y(0) = 1,
  D(y)(0) = 3}, y(x));
```

$$y(x) = \frac{6}{5}\, e^{2x} - \frac{1}{5}\, e^{-3x} \qquad \textbf{(4.5)}$$

Boundary Value Problem

A BVP has conditions of the form $y(x_0) = y_0$ and $y(x_1) = y_1$. Let's solve the boundary value problem $y'' + y' - 6y = 0, y(0) = 1, y(1) = 3$.

We already know the general solution is $y(x) = c_1 e^{2x} + c_2 e^{-3x}$ where c_1, c_2 are constants. The conditions allow us to find c_1, c_2. I am going to call the constants **_C1** and **_C2** so that it matches with the notation Maple gives us when using **dsolve**.

```
> y_original := _C1*exp(2*x) + _C2*exp(-3*x);
```

$$y\_original := \_C1\, e^{2x} + \_C2\, e^{-3x} \qquad \textbf{(5.1)}$$

To use the conditions, we plug them our expression for $y(x)$ which is given as **y_original**. This gives us a system of equations to solve.

```
> eqn1      := 1 = eval(y_original, x=0);
  eqn2      := 3 = eval(y_original, x=1);
  constants := solve([eqn1, eqn2], {_C1, _C2});
```

$$eqn1 := 1 = \_C1 + \_C2$$

$$eqn2 := 3 = \_C1\, e^{2} + \_C2\, e^{-3}$$

$$constants := \left\{\_C1 = -\frac{e^{-3} - 3}{e^{2} - e^{-3}}, \_C2 = \frac{-3 + e^{2}}{e^{2} - e^{-3}}\right\} \qquad (5.2)$$

Now that we know the constants `_C1` and `_C2`, we can substitute them back in to our original expression.

```
> eval(y_original, constants);
```

$$-\frac{\left(e^{-3} - 3\right) e^{2x}}{e^{2} - e^{-3}} + \frac{\left(-3 + e^{2}\right) e^{-3x}}{e^{2} - e^{-3}} \qquad (5.3)$$

This is the same solution that `dsolve` gives us.

```
> dsolve({diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = 0, y(0) = 1,
  y(1)=3}, y(x));
```

$$y(x) = -\frac{\left(e^{-3} - 3\right) e^{2x}}{e^{2} - e^{-3}} + \frac{\left(-3 + e^{2}\right) e^{-3x}}{e^{2} - e^{-3}} \qquad (5.4)$$

Homework

Do problem 21 in Section 7.7 Solve the differential equation $y'' + 16\,y = 0, y\left(\frac{\pi}{4}\right) = -3, y'\left(\frac{\pi}{4}\right) = 4$.